Maple 14における記号処理能力の強化

Maple 14では次の分野でその記号処理能力が改良されています。

assuming

- さまざまな前提条件下で結果を調べるにあたり関連した追加的柔軟性を与えられることにより、assuming コマンドは「この変数のみについて (only for this variable)」の形式の前提条件と「すべての変数について (for all variables)」の形式の前提条件が処理できます。また、Maple 14では assuming は方程式ラベルも処理できるようになっています。

```
> sqrt((y-x)^2);
```

$$\sqrt{(y-x)^2} \tag{1.1}$$

```
> simplify((1.1)) assuming positive and x < y;
```

$$y-x \tag{1.2}$$

以下は、もう少し複雑な例です。

```
> [about(x), about(y)] assuming x, posint, odd, x < 10, y, negint, even, y > -10;
Originally x, renamed x~:
  is assumed to be: AndProp(integer,LinearProp(2,integer,1),RealRange(1,
9))

Originally y, renamed y~:
  is assumed to be: AndProp(integer,LinearProp(2,integer,0),RealRange
(-9,-1))
```

$$[\,] \tag{1.3}$$

x および y に対する前提条件は
$x::(\mathrm{And}(posint, odd)), x < 10, y::(\mathrm{And}(negint, even)), -10 < y$ と同等で、従来どおり、一時的な前提条件です。つまり、assuming なしでは about(x) についても about(y) についてもなにも分かりません。

- assuming コマンドは Matrix と関連コマンドの任意グループ引数に対して使用できるようになりました。説明のため、この例では任意引数 readonly = true を使用します。

```
> simplify(Matrix([[sqrt(x^2), 0], [0, sqrt(y^2)]], readonly = true)) assuming positive;
```

$$\begin{bmatrix} x & 0 \\ 0 & y \end{bmatrix} \tag{1.4}$$

assuming コマンドは、最新の keyword parameters アプローチを利用した前提条件のオプションを採用し、実装しているすべての Maple マッピングおよびプログラムで一律に適用されることがなくなりました。例として次の 2 つの結果を比較します。

```
> a*x^2+b*x+c;
```

$$a x^2 + b x + c \tag{1.5}$$

```
> SolveTools:-Polynomial((1.5), x, domain = parametric);
```

(1.6)

$$\begin{cases} \begin{cases} \begin{cases} [x] & c=0 \\ [\ ] & c \neq 0 \end{cases} & b=0 \\ \left[-\frac{c}{b}\right] & b \neq 0 \end{cases} & a=0 \\ \left[\frac{1}{2}\,\frac{-b+\sqrt{b^2-4\,a\,c}}{a}, -\frac{1}{2}\,\frac{b+\sqrt{b^2-4\,a\,c}}{a}\right] & a \neq 0 \end{cases} \tag{1.6}$$

```
> SolveTools:-Polynomial((1.5), x, domain = parametric) assuming Not(0);
```

$$\left[\frac{1}{2}\,\frac{-b+\sqrt{b^2-4\,a\,c}}{a}, -\frac{1}{2}\,\frac{b+\sqrt{b^2-4\,a\,c}}{a}\right] \tag{1.7}$$

- 次のような map や @ を含む構文に対しても assuming が使用できるようになりました。

```
> [(y^2-2*y*x+x^2)^(1/2), (y^2+2*y*x+x^2)^(1/2)];
```

$$\left[\sqrt{y^2-2\,y\,x+x^2}, \sqrt{y^2+2\,y\,x+x^2}\right] \tag{1.8}$$

```
> map(simplify@factor, (1.8)) assuming real, y > x;
```

$$[y-x, |x+y|] \tag{1.9}$$

convert と FunctionAdvisor

- より多くの式が数学関数に変換できるようになりました。

```
> convert(AngerJ(a,z), Bessel) assuming a::integer;
```

$$\mathrm{BesselJ}(a, z) \tag{2.1}$$

- FunctionAdvisor は自動的に新しい式の恩恵を受けます。

```
> FunctionAdvisor(specialize,arctan(y,x), arccos);
```

$$\Bigg[\arctan(y, x) \tag{2.2}$$

$$= \frac{1}{x+\mathrm{I}\,y}\left(\mathrm{I}\sqrt{-\frac{(x+\mathrm{I}\,y)^2}{x^2+y^2}}\,\sqrt{x^2+y^2}\,\sqrt{\frac{\frac{x+\mathrm{I}\,y}{\sqrt{x^2+y^2}}-1}{1+\frac{x+\mathrm{I}\,y}{\sqrt{x^2+y^2}}}}\right.$$

$$\left.\sqrt{\frac{1+\frac{x+\mathrm{I}\,y}{\sqrt{x^2+y^2}}}{\frac{x+\mathrm{I}\,y}{\sqrt{x^2+y^2}}-1}}\,\arccos\left(\frac{x}{\sqrt{x^2+y^2}}\right)\right)+\pi$$

$$-\frac{\pi\sqrt{-1-\frac{x+\mathrm{I}\,y}{\sqrt{x^2+y^2}}}\ \sqrt{\frac{x+\mathrm{I}\,y}{\sqrt{x^2+y^2}}}}{\sqrt{-\frac{x+\mathrm{I}\,y}{\sqrt{x^2+y^2}}}\ \sqrt{1+\frac{x+\mathrm{I}\,y}{\sqrt{x^2+y^2}}}},$$

MathematicalFunctions:-with no restrictions on (x, y)

記号 _n_ についての _d^n/dz^n_ 微分

Maple 14 ではより多くの種類の記号階の微分方程式が扱えるようになりました。

- 累乗

```
> (a+z)^lambda;
```

$$(a+z)^{\lambda} \tag{3.1}$$

```
> %diff((3.1), z$n) = diff((3.1), z$n);
```

$$\frac{\partial^n}{\partial z^n}\,(a+z)^{\lambda}=\mathrm{pochhammer}(\lambda-n+1,n)\ (a+z)^{\lambda-n} \tag{3.2}$$

```
> 1/(a+b*z^3);
```

$$\frac{1}{a+b\,z^3} \tag{3.3}$$

```
> %diff((3.3), z$n) = diff((3.3), z$n);
```

$$\frac{\partial^n}{\partial z^n}\left(\frac{1}{a+b\,z^3}\right)=\sum_{\_\alpha=RootOf(a+b\_Z^3)}\frac{1}{3}\ \frac{\mathrm{pochhammer}(-n,n)\ (z-\_\alpha)^{-1-n}}{b\,\_\alpha^2} \tag{3.4}$$

- 初等関数

```
> tan(z);
```

$$\tan(z) \tag{3.5}$$

```
> %diff((3.5), z$n) = diff((3.5), z$n);
```

$$\frac{d^n}{dz^n}\tan(z)=-\mathrm{I}^{n+1}\,2^n\left(\mathrm{I}\tan(z)-1+\left(\left\{\begin{array}{ll}1 & n=0\\ 0 & otherwise\end{array}\right)\right)\left(\sum_{\_k1=0}^{n}\frac{(-1)^{-\_k1}\,\_k1!\ Stirling2(n,\_k1)\ (\mathrm{I}\tan(z)+1)^{-\_k1}}{2^{-\_k1}}\right)\right) \tag{3.6}$$

```
> csch(z);
```

$$\mathrm{csch}(z) \tag{3.7}$$

```
> %diff((3.7), z$n) = diff((3.7), z$n);
```

$$\frac{d^n}{dz^n}\mathrm{csch}(z)=\left(\mathrm{I}^n\right)^2\mathrm{csch}(z)\left(\sum_{\_j1=0}^{n}\left(\sum_{\_k1=0}^{\_j1}(-1)^{-\_k1}\,\mathrm{binomial}(n,\_j1)\ 2^{\_j1-\_k1}\,\_k1!\ Stirling2(\_j1,\_k1)\ (1-\coth(z))^{-\_k1}\right)\right) \tag{3.8}$$

- 特殊ベッセル (Bessel) 関数

```
> AngerJ(n,z);
```

$$\mathrm{AngerJ}(n, z) \tag{3.9}$$

> %diff((3.9), z$n) = diff((3.9), z$n);

$$\frac{\partial^n}{\partial z^n} \mathrm{AngerJ}(n, z) = \frac{\sum_{\_k1=0}^{n} (-1)^{\_k1} \mathrm{binomial}(n, \_k1) \mathrm{BesselJ}(2\_k1, z)}{2^n} \tag{3.10}$$

> HankelH1(n,z);

$$\mathrm{HankelH1}(n, z) \tag{3.11}$$

> %diff((3.11), z$n) = diff((3.11), z$n);

$$\frac{\partial^n}{\partial z^n} \mathrm{HankelH1}(n, z) = \frac{\sum_{\_k1=0}^{n} (-1)^{\_k1} \mathrm{binomial}(n, \_k1) \mathrm{BesselJ}(2\_k1, z)}{2^n} + \frac{I \left(\sum_{\_k1=0}^{n} (-1)^{\_k1} \mathrm{binomial}(n, \_k1) \mathrm{BesselY}(2\_k1, z) \right)}{2^n} \tag{3.12}$$

- **合流型超幾何関数**

> CylinderD(a,z);

$$\mathrm{CylinderD}(a, z) \tag{3.13}$$

> %diff((3.13), z$n) = diff((3.13), z$n);

$$\frac{\partial^n}{\partial z^n} \mathrm{CylinderD}(a, z) = \sqrt{\pi} \left(\sum_{\_k1=0}^{n} \mathrm{binomial}(n, \_k1) \left(\frac{d^{\_k1}}{dz^{\_k1}} \left(\frac{1}{e^{\frac{1}{4} z^2}} \right) \right) \right) \Bigg(- \frac{1}{\Gamma\left(-\frac{1}{2} a\right)} \Bigg(2^{\frac{1}{2} + \frac{1}{2} a} \Bigg(\sum_{\_k2=0}^{n - \_k1} \mathrm{binomial}(n - \_k1, \_k2) \mathrm{pochhammer}(2 - \_k2, \_k2) z^{1 - \_k2} \mathrm{pochhammer}(1 - n + \_k1 + \_k2, n - \_k1 - \_k2) z^{-n + \_k1 + \_k2} \mathrm{hypergeom}\left(\left[\frac{1}{2}, 1, -\frac{1}{2} a + \frac{1}{2} \right], \left[\frac{3}{2}, \frac{1}{2} \_k2 + 1 - \frac{1}{2} n + \frac{1}{2} \_k1, \frac{1}{2} - \frac{1}{2} n + \frac{1}{2} \_k1 + \frac{1}{2} \_k2 \right], \frac{1}{2} z^2 \right) \Bigg) \Bigg) + \frac{1}{\Gamma\left(-\frac{1}{2} a + \frac{1}{2}\right)} \Bigg(2^{\frac{1}{2} a} \mathrm{pochhammer}(-n + \_k1 + 1, n - \_k1) z^{-n + \_k1} \mathrm{hypergeom}\left(\left[1, -\frac{1}{2} a \right], \left[1 - \frac{1}{2} n + \frac{1}{2} \_k1, -\frac{1}{2} n + \frac{1}{2} \_k1 \right. \tag{3.14}$$

$$+\frac{1}{2}\Big], \frac{1}{2} z^2\Big)\Big)\Big)\Big)$$

新しい Solve コマンド

- Solve は PDEtools パッケージに新しく追加された統合コマンドで、ユーザーの入力に基づいて、解く対象が不等式、初期値または境界条件を含むのか含まないのか、そして連立代数方程式なのか、または連立微分方程式なのかによって solve、dsolve または pdsolve を呼び出します。さらに、Solve は非多項式型の方程式系についても指定変数に依存しない解を計算できる場合があります。

```
> with(PDEtools, Solve);
```

$$[Solve] \tag{4.1}$$

次は、微分方程式ではない式の例です。

```
> eq[1] := a*x^2 + b*x + c;
```

$$eq_1 := a\,x^2 + b\,x + c \tag{4.2}$$

```
> Solve(eq[1], x);
```

$$x = \frac{1}{2}\,\frac{-b + \sqrt{b^2 - 4\,a\,c}}{a},\ x = -\frac{1}{2}\,\frac{b + \sqrt{b^2 - 4\,a\,c}}{a} \tag{4.3}$$

次は、ODE 問題とその級数解の例です。

```
> eq[2] := diff(y(x),x) = y(x);
```

$$eq_2 := \frac{\mathrm{d}}{\mathrm{d}x}\,y(x) = y(x) \tag{4.4}$$

```
> Solve(eq[2], series);
```

$$y(x) = y(0) + y(0)\,x + \frac{1}{2}\,y(0)\,x^2 + \frac{1}{6}\,y(0)\,x^3 + \frac{1}{24}\,y(0)\,x^4 + \frac{1}{120}\,y(0)\,x^5 + \mathrm{O}(x^6) \tag{4.5}$$

次は、境界条件付きの PDE 問題の例です。

```
> eq[3] := [diff(u(x, t), t)+c*(diff(u(x, t), x)) = -lambda*u(x, t), u(x, 0) = phi(x)];
```

$$eq_3 := \left[\frac{\partial}{\partial t}\,u(x,t) + c\left(\frac{\partial}{\partial x}\,u(x,t)\right) = -\lambda\,u(x,t),\ u(x,0) = \phi(x)\right] \tag{4.6}$$

```
> Solve(eq[3]);
```

$$u(x,t) = \phi(x - t\,c)\,\mathrm{e}^{-\lambda t} \tag{4.7}$$

上記例は solve、dsolve または pdsolve を 1 つのコマンドで呼び出し、入力および出力に対して統一された形式を持つ利点を示しています。Solve は追加機能も提供します：与えられた変数に依存しない解を計算できます。independentof が指定されていると、不等式が含まれていても系を解くことができます。

```
> eq[4] := [k*a*c*(a+b)*exp(k*d*t)-2*a*exp(k*t)*k+Q*(-c+a)*x, a <> 0];
```

$$eq_4 := \left[k\,a\,c\,(a+b)\,\mathrm{e}^{k\,d\,t} - 2\,a\,\mathrm{e}^{k\,t}\,k + Q\,(-c+a)\,x,\ a \neq 0\right] \tag{4.8}$$

```
> Solve(eq[4], {a, b, c, d}, independentof = {t, x});
```

$$\left\{a = a,\ b = -\frac{a^2 - 2}{a},\ c = a,\ d = 1\right\} \tag{4.9}$$

微分方程式または微分方程式系についても与えられた変数と独立した解を計算できます。

```
> eq[5] := diff(f(x,y),x)*diff(g(x,y),x) + diff(f(x,y),y)*diff(g(x,y),y) + g(x,y)*(diff(f(x,y), x,x) + diff(f(x,y), y,y)) = -1;
```

(4.10)

$$eq_5 := \left(\frac{\partial}{\partial x} f(x,y)\right)\left(\frac{\partial}{\partial x} g(x,y)\right) + \left(\frac{\partial}{\partial y} f(x,y)\right)\left(\frac{\partial}{\partial y} g(x,y)\right) + g(x,y)\left(\frac{\partial^2}{\partial x^2} f(x,y) + \frac{\partial^2}{\partial y^2} f(x,y)\right) = -1 \tag{4.10}$$

次は、x にも、y にも依存しない PDE の解の例です。

```
> Solve(eq[5], independentof = x);
```

$$\left\{ f(x,y) = \_F1(y), g(x,y) = \frac{-y + \_C1}{\frac{d}{dy}\_F1(y)} \right\} \tag{4.11}$$

```
> Solve(eq[5], independentof = y);
```

$$\left\{ f(x,y) = \_F1(x), g(x,y) = \frac{-x + \_C1}{\frac{d}{dx}\_F1(x)} \right\} \tag{4.12}$$

代数ソルバ

- SolveTools[Polynomial] コマンドと多くの新規オプションにより、SolveTools パッケージの多項式ソルバが利用しやすくなりました。新規コマンド SolveTools[Polynomial] を使用すると、1 変数多項式ソルバを直接利用することができ、SolveTools[PolynomialSystem] を使用すると、多変数多項式ソルバを直接利用することができます。純粋な多項式に関しては、多くの前処理を省略し、SolveTools[Linear] 並みのオーバーヘッドしか送らないこれらのコマンドは solve より効率的です。

```
> with(SolveTools):
> f1 := expand((x-1)^4*eval(z^4-z-1,z=x^3+x)):
> Polynomial(f1, x, domain=integer);
```

$$[1, 1, 1, 1] \tag{5.1}$$

```
> Polynomial(f1, x, domain=real);
```

$$[RootOf(\_Z^{12} + 4\_Z^{10} + 6\_Z^8 + 4\_Z^6 + \_Z^4 - \_Z^3 - \_Z - 1, -0.5542396981), RootOf(\_Z^{12} + 4\_Z^{10} + 6\_Z^8 + 4\_Z^6 + \_Z^4 - \_Z^3 - \_Z - 1, 0.7679130647), 1, 1, 1, 1] \tag{5.2}$$

- 次は、SolveTools[Polynomial] を使用してパラメトリック多項式の区分的解を求める例です。

```
> Polynomial(a*x^2-(b+a)*x+b, x, domain=parametric);
```

$$\begin{cases} \begin{cases} [x] & -b = 0 \\ [1] & -b \neq 0 \end{cases} & a = 0 \\ \left[\frac{b}{a}, 1\right] & a \neq 0 \end{cases} \tag{5.3}$$

```
> PolynomialSystem({y^3+1,y+x^2-1}, {x,y}, domain=real);
```

$$\{x = RootOf(\_Z^2 - 2, -1.414213562), y = -1\}, \{x = RootOf(\_Z^2 - 2, 1.414213562), y = -1\} \tag{5.4}$$

```
> PolynomialSystem({y^3+1,y+x^2-1}, {x,y}, domain=real, explicit);
```

$$\{x = -\sqrt{2}, y = -1\}, \{x = \sqrt{2}, y = -1\} \tag{5.5}$$

```
> PolynomialSystem({y-x^2+1, y+x^2-1}, [x,y], engine=groebner, backsubstitute=false);
```

$$[\{y, x-1\}, \{\ \}], [\{y, x+1\}, \{\ \}] \tag{5.6}$$

• SolveTools[Engine] コマンドによって、solve に新しい下位インターフェイスが追加されました。このインターフェイスでは最低限の前処理や後処理と、ユーザープログラム内での使用により適した統一形式の入力および出力が使用されます。solve はまだ Maple の求解機能への推奨ユーザーレベルインターフェイスです。

```
> Engine({x^2+1},{x});
```

$$[\{x=-I\}, \{x=I\}] \tag{5.7}$$

• allvalues コマンドは非多項式型の RootFinding 入力に対して RootOf ライブラリを使用し、より分かりやすい解が返せるように変更されました。

```
> solve(x*sin(x)-1);
```

$$RootOf(\_Z\sin(\_Z)-1) \tag{5.8}$$

```
> allvalues((5.8)) ;
```

$$RootOf(\_Z\sin(\_Z)-1, -1.114157141), RootOf(\_Z\sin(\_Z)-1, -2.772604708), RootOf(\_Z\sin(\_Z)-1, 1.114157141), RootOf(\_Z\sin(\_Z)-1, 2.772604708) \tag{5.9}$$

新しい IntegrationTools コマンド

• IntegrationTools パッケージには ExpandMultiple コマンドと CollapseNested コマンドが追加されました。これらのコマンドは受動的積分表現を別の受動的積分表現に変換することができます。

```
> with(IntegrationTools):
> i1 := Int(Int(f(x,y),x),y);
```

$$i1 := \int\int f(x,y)\,dx\,dy \tag{6.1}$$

```
> i2 := CollapseNested( i1 );
```

$$i2 := \int\int f(x,y)\,dx\,dy \tag{6.2}$$

```
> lprint(i2=i1);
Int(f(x,y),[x, y]) = Int(Int(f(x,y),x),y)
> i3 := ExpandMultiple( i2 );
```

$$i3 := \int\int f(x,y)\,dx\,dy \tag{6.3}$$

```
> lprint(i3=i2);
Int(Int(f(x,y),x),y) = Int(f(x,y),[x, y])
```

不定積分と定積分

• 楕円関数の不定積分が改良されました。

```
> %int(x*(EllipticK(2/x)*(1-2/x^2)-EllipticE(2/x)), x);
```

$$\int x\left(\mathrm{EllipticK}\left(\frac{2}{x}\right)\left(1-\frac{2}{x^2}\right)-\mathrm{EllipticE}\left(\frac{2}{x}\right)\right)dx \tag{7.1}$$

```
> value((7.1)) ;
```

$$-\frac{1}{2}\,\mathrm{EllipticE}\left(\frac{2}{x}\right)x^2+\frac{1}{2}\,\mathrm{EllipticK}\left(\frac{2}{x}\right)x^2-2\,\mathrm{EllipticK}\left(\frac{2}{x}\right) \tag{7.2}$$

この改良により、以下のような被積分関数についても定積分が計算できるようになりました。

```
> subsop(2 = (x = alpha..infinity), (7.1)) ;
```

$$\int_{\alpha}^{\infty} x\left(\mathrm{EllipticK}\left(\frac{2}{x}\right)\left(1-\frac{2}{x^2}\right)-\mathrm{EllipticE}\left(\frac{2}{x}\right)\right)dx \tag{7.3}$$

```
> value((7.3)) ;
```

(7.4)

$$\frac{1}{2}\,\mathrm{EllipticE}\left(\frac{2}{\alpha}\right)\alpha^2-\frac{1}{2}\,\mathrm{EllipticK}\left(\frac{2}{\alpha}\right)\alpha^2+2\,\mathrm{EllipticK}\left(\frac{2}{\alpha}\right)-\frac{1}{2}\,\pi \tag{7.4}$$

- 定積分器は、微分積分学の基本定理を適用して極限を計算する際は別の実装を使用するように改良されました。これにより、以前の Maple バージョンでは未評価の極限が返されていたような場合でも閉形式の解が返せるようになりました。この新しいコードは、オプション method=FTOCMS を使用して直接利用することもできます。method オプションの詳細については、int/method を参照してください。

```
> int( (x^3+x^2)/(exp(x)-1), x=0..infinity);
```

$$\frac{1}{15}\,\pi^4+2\,\zeta(3) \tag{7.5}$$

```
> int( (x^3+x^2)/(exp(x)-1), x=0..infinity, method=FTOCMS);
```

$$\frac{1}{15}\,\pi^4+2\,\zeta(3) \tag{7.6}$$

```
> int( (x^3+x^2)/(exp(x)-1), x=0..infinity, method=NoFTOCMS);
```

$$-\frac{1}{15}\,\pi^4+2\,\zeta(3)+\lim_{x\to\infty-}\left(-\frac{1}{4}\,x^4+x^3\ln\left(1-\mathrm{e}^x\right)+3\,x^2\,\mathrm{polylog}\left(2,\mathrm{e}^x\right)-6\,x\,\mathrm{polylog}\left(3,\mathrm{e}^x\right)+6\,\mathrm{polylog}\left(4,\mathrm{e}^x\right)-\frac{1}{3}\,x^3+x^2\ln\left(1-\mathrm{e}^x\right)+2\,x\,\mathrm{polylog}\left(2,\mathrm{e}^x\right)-2\,\mathrm{polylog}\left(3,\mathrm{e}^x\right)\right) \tag{7.7}$$

関連項目

Index of New Maple 14 Features